Лабораторная работа № 1.2
Структура окна *Maple*. Арифметические операции, числа, константы и стандартные функции. Элементарные преобразования математических выражений. Функции в *Maple*. Операции оценивания. Решение уравнений и систем.

Часть первая

§1. Структура окна *Maple*

Maple – это пакет для аналитических вычислений на компьютере, содержащий более двух тысяч команд, которые позволяют решать задачи алгебры, геометрии, математического анализа, дифференциальных уравнений, статистики, математической физики.

Для того, чтобы запустить *Maple*, необходимо в *Главном меню Windows* выбрать в группе *Программы* название данного приложения: *Maple*.

Maple представляет собой типичное окно *Windows*, которое состоит из *Строки названия*, *Основного меню*, *Панели инструментов*, *Рабочего поля* и *Строки состояния*, а также *Линейки* и *Полос прокрутки*.

Вид фрагмента окна *Maple* 6, содержащего *Строку названия, Основное меню, Панель инструментов*:

Пункты *Основного меню*:

File (Файл) – содержит стандартный набор команд для работы с файлами, например: сохранить файл, открыть файл, создать новый файл и т.д.

Edit (Правка) – содержит стандартный набор команд для редактирования текста, например: копирование, удаление выделенного текста в буфер обмена, отмена команды и т.д.

View (Вид) – содержит стандартный набор команд, управляющих структурой окна *Maple*.

Insert (Вставка) – служит для вставки полей разных типов: математических текстовых строк, графических двух и трехмерных изображений.

Format (Формат) – содержит команды оформления документа, например: установка типа, размера и стиля шрифта.

Options (Параметры) – служит для установки различных параметров ввода и вывода информации на экран, принтер, например, таких как качество печати.

Windows (Окно) – служит для перехода из одного рабочего листа в другой.

Help (Справка) – содержит подробную справочную информацию о *Maple*.

Работа в *Maple* проходит в режиме сессии – пользователь вводит предложения (команды, выражения, процедуры), которые воспринимаются условно и обрабатываются *Maple*. Рабочее поле разделяется на три части:

1) область ввода - состоит из командных строк. Каждая командная строка начинается с символа >;
2) область вывода - содержит результаты обработки введенных команд в виде аналитических выражений, графических объектов или сообщений об ошибке;
3) область текстовых комментариев - содержит любую текстовую информацию, которая может пояснить выполняемые процедуры. Текстовые строки не воспринимаются *Maple* и никак не обрабатываются.

Для того, чтобы переключить командную строку в текстовую, следует на *Панели инструментов* нажать мышью на кнопку



Обратное переключение текстовой строки в командную осуществляется нажатием на *Панели инструментов* на кнопку



Задание 1.

1. Запустите *Maple*.
2. После запуска *Maple* первая строка оказывается командной. Переведите ее в текстовую. Наберите в этой строке: «Лабораторная работа №1» и название темы. Перейдите на следующую строку, нажав *Enter*.
3. В новой строке наберите «Выполнил студент » и свою фамилию. Нажмите *Enter*.
4. На следующей строке наберите «Задание №1».
5. Сохраните свой файл на дискете. Для этого в меню **Fail** выберите пункт **Save** и наберите имя вашего файла в виде: Фамилия_1, где указывается ваша фамилия и 1 – номер лабораторной работы.
6. После этого в следующей строке наберите текст: «Файл с заданиями лабораторной работы №1 сохранен под именем: Фамилия_N».

В дальнейшем выполнение каждой лабораторной работы должно оформляться таким способом. В начале каждой лабораторной работы следует набирать текст: «Лабораторная работа N», N – номер темы. Выполнение каждого задания следует начинать с текстового комментария: «Задание N». Для правильности вычислений перед выполнением каждого пункта задания следует выполнять команду **restart**. Перед выполнением контрольных заданий следует набирать в текстовом режиме «Контрольные задания». После окончания выполнения работы необходимо сохранить файл со всеми выполненными заданиями на диск. Имя вашего файла набирается в виде: Фамилия_N, где указывается ваша фамилия и N – номер темы.

§2. Арифметические операции.
Целые и рациональные числа, константы в *Maple*

Математические константы и арифметические операции.

Основные математические константы:

Pi – число π; **I** – мнимая единица i; **infinity** – бесконечность; **Gamma** – константа Эйлера; **true**, **false** – логические константы, обозначающие истинность и ложность высказывания.

Знаки арифметических операций:

+ - сложение; **–** - вычитание;

***** - умножение; **/** - деление;

^ - возведение в степень; **!** – факториал.

Знаки сравнения: **<, >, >=,<=, <>, =.**

Комплексные, целые и рациональные числа.

Числа в *Maple* бывают действительные (real) и комплексные (complex). Комплексное число записывается в алгебраической форме $z=x+iy$, и в командной строке такая запись должна выглядеть так:

> z:=x+I*y;

Вещественные числа разделяются на целые и рациональные. Целые числа (integer) выражаются цифрами в десятичной записи. Рациональные числа могут быть представлены в 3-х видах:

1) рациональной дроби с использованием оператора деления, например: **28/70**;
2) с плавающей запятой (float), например: **2.3;**
3) в показательной форме, например: **1,602*10^(-19)** означает $1{,}602\cdot 10^{-19}$.

Для того, чтобы получить рациональное число не в точной форме, а в виде приближенного значения (числа с плавающей запятой), следует дописывать к целой части числа .0. Пример:

> **75/4;**

$$\frac{75}{4}$$

> 75/4.0;

$$18.75000000$$

В *Maple* можно записать буквы греческого алфавита в полиграфическом виде. Для этого в командной строке набирается название греческой буквы. Например, буква α получится, если набрать **alpha**,

β **- beta ,** ε **- epsilon,** η **- eta,** θ **- theta,** λ **- lambda,** ξ **-xi,** χ **- chi.**

Заглавные греческие буквы можно записать, если набирать название греческой буквы с заглавной, например, чтобы получить Ω, следует набрать **Omega**. Греческие буквы также можно набирать с помощью специального меню.

Задание 2.

1. Перейдите в текстовый режим и наберите «Задание №2». После не забудьте перейти в режим командной строки.
2. Вычислите значение $\dfrac{\sqrt{6+2\sqrt{5}}-\sqrt{6-2\sqrt{5}}}{\sqrt{3}}$. Для этого в командной строке наберите:

> (sqrt(6+2*sqrt(5))-sqrt(6-2*sqrt(5)))/sqrt(3);

и нажмите *Enter*. В результате получится точное значение $\frac{2}{3}\sqrt{3}$.

3. Наберите формулы $\omega=\dfrac{\theta}{t}$ и $|f(x)-\delta|<\varepsilon$. Для этого в командной строке наберите:

> **omega=theta/t; abs(f(x)-delta)<epsilon;**

нажмите *Enter*.

§3. Синтаксис команд. Стандартные функции

Синтаксис команд.

Стандартная команда *Maple* состоит из имени команды и ее параметров, указанных в круглых скобках: **command(p1, p2, …)**. В конце каждой команды должен быть знак (**;**) или (**:**). Разделитель (**;**) означает, что в области вывода после выполнения этой команды будет сразу виден результат. Разделитель (**:**) используется для отмены вывода, то есть когда команда выполняется, но ее результат на экран не выводится.

Символ процента (**%**) служит для вызова предыдущей команды. Этот символ играет роль краткосрочной замены предыдущей команды с целью сокращения записи. Пример использования (**%**):

> **a+b;**

$$a+b$$

> **%+c;**

$$a+b+c.$$

Для присвоения переменной заданного значения используется знак присвоить (**:=**).

Когда программа *Maple* запускается, она не имеет ни одной команды, полностью загруженной в память. Большая часть команд имеют указатели их нахождения, и при вызове они загружаются автоматически. Другие команды находятся в стандартной библиотеке и перед выполнением обязательно должны быть вызваны командой **readlib(command)**, где **command** – имя вызываемой команды. Остальная часть процедур *Maple* содержится в специальных библиотеках подпрограмм, называемых пакетами. Пакеты необходимо подгружать при каждом запуске файла с командами из этих библиотек. Имеется два способа вызова команды из пакета:

1) можно загрузить весь пакет командой **with(package)** где **package** – имя пакета;
2) вызов какой-нибудь одной команды **command** из любого пакета **package** можно осуществить, если набрать команду в специальном формате:

   > **package[command](options);**

где вначале записывается название пакета **package**, из которого надо вызвать команду, а затем в квадратных скобках набирается имя самой команды **command**, и после чего в круглых скобках следуют параметры **options** данной команды.

К библиотекам подпрограмм *Maple* относятся, например, следующие пакеты: **linalg** – содержит операции линейной алгебры; **stats** – для решения задач статистической обработки массивов данных; **student** – содержит команды, позволяющие провести поэтапное решение задачи в аналитическом виде с промежуточными вычислениями.

Стандартные функции.

Стандартные функции *Maple*	
Математическая запись	Запись в *Maple*
e^x	**exp(x)**
$\ln x$	**ln(x)**
$\lg x$	**log10(x)**
$\log_a x$	**log[a](x)**
$\sqrt{x}$	**sqrt(x)**
$\lvert x \rvert$	**abs(x)**
$\sin x$	**sin(x)**
$\cos x$	**cos(x)**
$\mathrm{tg} x$	**tan(x)**
$\mathrm{ctg} x$	**cot(x)**
$\sec x$	**sec(x)**
$\mathrm{cosec} x$	**csc(x)**
$\arcsin x$	**arcsin(x)**
$\arccos x$	**arccos(x)**
$\mathrm{arctg} x$	**arctan(x)**
$\mathrm{arcctg} x$	**arccot(x)**
$\mathrm{sh} x$	**sinh(x)**
$\mathrm{ch} x$	**cosh(x)**
$\mathrm{th} x$	**tanh(x)**
$\mathrm{cth} x$	**coth(x)**
$\delta(x)$ - функция Дирака	**Dirac(x)**
$\theta(x)$ - функция Хевиссайда	**Heaviside(x)**

Maple содержит огромное количество специальных функций, таких, как Бесселевы функции, Эйлеровы бета- и гамма – функции, интеграл ошибок, эллиптические интегралы, различные ортогональные полиномы.

С помощью функции **exp(x)** определяется число *e*=2.718281828… посредством записи **exp(1)**.

Тождественные преобразования выражений.

Раскрытие скобок выражения **eq** осуществляется командой **expand(eq)**. Пример:

> eq:=(x+1)*(x-1)*(x^2-x+1)*(x^2+x+1);

$$eq := (x+1)(x-1)(x^2-x+1)(x^2+x+1)$$

> expand(eq);

$$x^6 - 1$$

Разложение многочлена на множители осуществляется командой **factor(eq).**

Пример:

> **p:=x^5-x^4-7*x^3+x^2+6*x;**

$$p := x^5 - x^4 - 7x^3 + x^2 + 6x$$

> **factor(p);**

$$x(x-1)(x-3)(x+2)(x+1)$$

Команда **expand** может иметь дополнительный параметр, позволяющий при раскрытии скобок оставлять определенное выражение без изменений. Например, пусть требуется каждое слагаемое выражения $\ln x + e^x - y^2$ умножить на выражение $(x+a)$. Тогда в командной строке следует написать:

> **expand((x+a)*(ln(x)+exp(x)-y^2), (x+a));**

$$(x+a)\ln x + (x+a)e^x - (x+a)y^2$$

Дробь можно привести к нормальному виду с помощью команды **normal(eq)**. Например:

> **f:=(a^4-b^4)/((a^2+b^2)*a*b);**

$$f := \frac{a^4 - b^4}{(a^2+b^2)ab}$$

> **normal(f);**

$$\frac{a^2 - b^2}{ab}$$

Упрощение выражений осуществляется командой **simplify(eq)**.

Пример:

> **eq:=(cos(x)-sin(x))*(cos(x)+sin(x)):**
> **simplify(eq);**

$$2\cos(x)^2 - 1$$

Приведение подобных членов в выражении осуществляется командой **collect(exp,var),** где **exp** – выражение, **var** – имя переменной, относительно которой следует собирать подобные. В команде **simplify** в качестве параметров можно указать, какие выражения преобразовывать. Например, при указании **simplify(eq,trig)** будет производиться упрощение при использовании большого числа тригонометрических соотношений. Стандартные параметры имеют названия: **power** – для степенных преобразований; **radical** или **sqrt** – для преобразования корней; **exp** – преобразование экспонент; **ln** – преобразование логарифмов. Использование параметров намного увеличивает эффективность команды **simplify**.

Объединить показатели степенных функций или понизить степень тригонометрических функций можно при помощи команды **combine(eq,param)**, где **eq** – выражение, **param** – параметры, указывающие, какой тип функций преобразовать, например, **trig** – для тригонометрических, **power** – для степенных. Пример:

> **combine(4*sin(x)^3, trig);**

$$-\sin(3x) + 3\sin(x)$$

Для упрощения выражений, содержащих не только квадратные корни, но и корни других степеней, лучше использовать команду **radnormal(eq)**. Пример:

> **sqrt(3+sqrt(3)+(10+6*sqrt(3))^(1/3))=**
> **radnormal(sqrt(3+sqrt(3)+(10+6*sqrt(3))^(1/3)));**

$$\sqrt{3 + \sqrt{3} + (10 + 6\sqrt{3})^{1/3}} = 1 + \sqrt{3}$$

С помощью команды **convert(exp, param)**, где **exp** – выражение, которое будет преобразовано в указанный тип **param**. В частности, можно преобразовать выражение, содержащее sin*x* и cos*x*, в выражение, содержащее только tg*x*, если указать в качестве параметра **tan**, или, наоборот, tg*x*, ctg*x* можно перевести в sin*x* и cos*x*, если в параметрах указать **sincos**.

Вообще, команда **convert** имеет более широкое назначение. Она осуществляет преобразование выражения одного типа в другой. Например: **convert(list, vector)** – преобразование некоторого списка **list** в вектор с теми же элементами; **convert(expr, string)** – преобразование математического выражения в его текстовую запись. Для вызова подробной информации о назначении параметров команды **convert** следует обратиться к справочной системе, набрав **convert[termin]**.

Если вы забыли параметры какой-либо команды, то можно воспользоваться справочной системой *Maple*. Для вызова справки по конкретной команде, следует выделить набранное имя этой команды и нажать клавишу F1. Если команда набрана правильно, то появится описание этой команды (в большинстве версий *Maple* помощь на английском языке).

Задание 3.

1. Перейдите в текстовый режим и наберите «Задание №3». После не забудьте перейти в режим командной строки. Перед выполнением каждого пункта этого задания обязательно набирайте команду обновления **restart;**
2. Разложить полином на множители $p = x^3 + 4x^2 + 2x - 4$. Для этого наберите в командной строке:

   > factor(x^3+4*x^2+2*x-4);

После нажатия клавиши *Enter* должно получиться $(x+2)(x^2+2x-2)$.

3. Упростить выражение $\frac{1+\sin 2x+\cos 2x}{1+\sin 2x-\cos 2x}$. Наберите:

> eq:=(1+sin(2*x)+cos(2*x))/(1+sin(2*x)-cos(2*x)):
> convert(eq, tan):
> eq=normal(%);

$$\frac{1+\sin(2x)+\cos(2x)}{1+\sin(2x)-\cos(2x)} = \frac{1}{\tan(x)}.$$

4. Упростить выражение $3(\sin^4 x + \cos^4 x) - 2(\sin^6 x + \cos^6 x)$. Для этого наберите:

> eq:=3*(sin(x)^4+cos(x)^4)-2*(sin(x)^6+cos(x)^6):
> eq=combine(eq, trig);

$$3\sin(x)^4 + 3\cos(x)^4 - 2\sin(x)^6 - 2\cos(x)^6 = 1$$

§4. Способы задания функций. Замена переменных

В *Maple* имеется несколько способов представления функции.

Способ 1. Определение функции с помощью оператора присваивания (**:=**): какому-то выражению присваивается имя, например:

> f:=sin(x)+cos(x);

$$f := \sin(x) + \cos(x)$$

Если задать конкретное значение переменной *x*, то получится значение функции *f* для этого *x*. Например, если продолжить предыдущий пример и вычислить значение **f** при $x = \pi/4$, то следует записать:

> x:=Pi/4;

$$x := \frac{\pi}{4}$$

> f;

$$\sqrt{2}$$

После выполнения этих команд переменная *x* имеет заданное значение $\pi/4$.

Чтобы насовсем не присваивать переменной конкретного значения, удобнее использовать команду подстановки **subs({x1=a1, x2=a2,…, },f),** где в фигурных скобках указываются переменные **xi** и их новые значения **ai** (*i*=1,2,…), которые следует подставить в функцию **f** . Например:

> f:=x*exp(-t);

$$f := xe^{(-t)}$$

> subs({x=2,t=1},f);

$$2e^{(-1)}$$

Все вычисления в *Maple* по умолчанию производятся символьно, то есть результат будет содержать в явном виде иррациональные константы, такие как, e, π и другие. Чтобы получить приближенное значение в виде числа с плавающей запятой, следует использовать команду **evalf(expr,t),** где **expr** – выражение, **t** – точность, выраженная в числах после запятой. Например, в продолжение предыдущего примера, вычислим полученное значение функции приближенно:

> evalf(%);

$$.7357588824$$

Здесь использован символ (**%**) для вызова предыдущей команды.

Способ 2. Определение функции с помощью функционального оператора, который ставит в соответствие набору переменных **(x1,x2,…)** одно или несколько выражений **(f1,f2,…)**. Например, определение функции двух переменных с помощью функционального оператора выглядит следующим образом:

> f:=(x,y)->sin(x+y);

$$f := \sin(x + y)$$

Обращение к этой функции осуществляется наиболее привычным в математике способом, когда в скобках вместо аргументов функции указываются конкретные значения переменных. В продолжение предыдущего примера вычисляется значение функции:

> f(Pi/2,0);

$$1$$

Способ 3. С помощью команды **unapply(expr,x1,x2,…)**, где **expr** – выражение, **x1,x2,…** – набор переменных, от которых оно зависит, можно преобразовать выражение **expr** в функциональный оператор. Например:

> f:=unapply(x^2+y^2,x,y);

$$f := (x, y) -> x^2 + y^2$$

➢ **f(-7,5);**

$$74$$

В *Maple* имеется возможность определения неэлементарных функций вида

$$f(x) = \begin{cases} f_1(x),\ x < a_1 \\ f_2(x),\ a_1 < x < a_2 \\ \ldots\ldots\ldots\ldots\ldots\ldots \\ f_n(x),\ x > a_n \end{cases}$$

посредством команды

> piecewise(cond_1,f1, cond_2, f2, …).

Например, функция

$$f(x) = \begin{cases} 0,\ x < 0 \\ x,\ 0 \le x < 1 \\ \sin x,\ x \ge 1 \end{cases}$$

записывается следующим образом:

> f:=piecewise(x<0, 0, 0<=x and x<1, x, x>=1, sin(x));

$$f := \begin{cases} 0 & x < 0 \\ x & -x \le 0 \text{ and } x - 1 < 0 \\ \sin x & 1 \le x \end{cases}$$

Задание 4.

Не забудьте, что выполнение всех последующих заданий должно начинаться с текстовой строки, содержащей «Задание №», где № – номер задания. Также помните, что для правильности вычислений перед выполнением каждого пункта задания следует выполнять команду **restart**. Перед выполнением контрольных заданий следует набирать в текстовом режиме «Контрольные задания». Эти правила оформления относятся ко всем лабораторным работам.

1. Определите функцию $f=\sqrt{1-x^2-y^2}$ и перейдите в ней к полярным координатам $x=\rho\cos\varphi$, $y=\rho\sin\varphi$. Упростите полученное выражение. Для этого наберите:

   > f:=sqrt(1-x^2-y^2);

$$f=\sqrt{1-x^2-y^2}$$

   > f:=subs({x=rho*cos(phi),y=rho*sin(phi)},f);

$$f=\sqrt{1-\rho^2\cos(\phi)^2-\rho^2\sin(\phi)^2}$$

   > f:=simplify(%);

$$f=\sqrt{1-\rho^2}$$

2. Определите функцию $f(x)=\begin{cases}x,\ x<-1\\ -x^2,\ -1\le x<1\\ -x,\ x\ge 1\end{cases}$ и прибавьте к ней x. Для этого наберите:

> f:=piecewise(x<-1, x, -1<=x and x<1, -x^2, x>=1, -x);

$$f:=\begin{cases}x & x<-1\\ -x^2 & -1-x\le 0 \text{ and } x-1<0\\ -x & 1\le x\end{cases}$$

> %+x: simplify(%);

$$\begin{cases}2x & x<-1\\ x-x^2 & x\le 1\\ 0 & 1<x\end{cases}$$

§5. Решение уравнений

Решение обыкновенных уравнений.

Для решения уравнений в *Maple* существует универсальная команда **solve(eq,x)**, где **eq** – уравнение, **x** – переменная, относительно которой уравнение надо разрешить. В результате выполнения этой команды в строке вывода появится выражение, которое является решением данного уравнения. Например:

> solve(a*x+b=c,x);

$$-\frac{b-c}{a}$$

Если уравнение имеет несколько решений, которые вам понадобятся для дальнейших расчетов, то команде **solve** следует присвоить какое-нибудь имя **name**. Обращение к какому-либо **k**–ому решению данного уравнения производится указанием его имени с номером решения **k** в квадратных скобках: **name[k]**. Например:

> x:=solve(x^2-a=0,x);

$$x:=-\sqrt{a},\sqrt{a}$$

> x[1];

$$-\sqrt{a}$$

> x[2];

$$\sqrt{a}$$

> x[1]+x[2];

Решение систем уравнений.

Системы уравнений решаются с помощью такой же команды **solve({eq1,eq2,…},{x1,x2,…})**, только теперь в параметрах команды следует указывать в первых фигурных скобках через запятую уравнения, а во вторых фигурных скобках перечисляются через запятую переменные, относительно которых требуется решить систему. Если вам будет необходимо для дальнейших вычислений использовать полученные решения уравнений, то команде **solve** следует присвоить какое-нибудь имя **name**. Затем выполняется присвоения команда **assign(name)**. После этого над решениями можно будет производить математические операции. Например:

> s:=solve({a*x-y=1,5*x+a*y=1},{x,y});

$$s:=\{\ x=\frac{a+1}{5+a^2},\ y=\frac{a-5}{5+a^2}\ \}$$

> assign(s); simplify(x-y);

$$\frac{6}{5+a^2}$$

Численное решение уравнений.

Для численного решения уравнений, в тех случаях, когда трансцендентные уравнения не имеют аналитических решений, используется специальная команда **fsolve(eq,x)**, параметры которой такие же, как и команды **solve**. Например:

> x:=fsolve(cos(x)=x,x);

$$x:=.7390851332$$

Задание 5.

1. Найти все решения системы уравнений $\begin{cases} x^2-y^2=1, \\ x^2+xy=2. \end{cases}$

   Наберите:
   > eq:={x^2-y^2=1,x^2+x*y=2};
   > _EnvExplicit:=true:
   > s:=solve(eq,{x,y});

$$s:=\{x=\frac{2}{3}\sqrt{3}, y=\frac{1}{3}\sqrt{3}\}, \{x=-\frac{2}{3}\sqrt{3}, y=-\frac{1}{3}\sqrt{3}\}$$

Теперь найдите сумму двух наборов решений. Наберите:

> x1:=subs(s[1],x): y1:=subs(s[1],y):
x2:=subs(s[2],x): y2:=subs(s[2],y):
> x1+x2; y1+y2;

Чему равны эти суммы решений?

2. Численно решите уравнение $x^2 = \cos(x)$. Наберите:

   > x=fsolve(x^2=cos(x),x);

$$x=.8241323123$$

3. Найдите функцию $f(x)$, удовлетворяющую уравнению $f^2(x)-2f(x)=x$. Наберите:

   > F:=solve(f(x)^2-2*f(x)=x,f);

F:= **proc**(x) RootOf(_Z^2–2*_Z–x) **end**

> f:=convert(F(x), radical);

$$f:=1+\sqrt{1+x}$$

1. Выполните все контрольные задания. Перед их выполнением не забудьте набрать в текстовом режиме «Контрольные задания». Результаты выполнения заданий покажите преподавателю.

2. Сохраните файл со всеми выполненными заданиями на диск.
3. Ответьте на все контрольные вопросы.

Контрольные задания.

1. Вычислить точное и приближенное значения выражения: $\mathrm{arctg}3 - \arcsin\frac{\sqrt{5}}{5}$.
2. Записать формулы: $\omega(k) = \alpha k^2 + \beta k^4$; $\xi = ae^{-\gamma r}\cos(\omega t + \varphi)$.
3. Разложить на множители полином $p = x^3 - 4x^2 + 5x - 2$.
4. Упростить выражение $\sin^2 3x - \sin^2 2x - \sin 5x \sin x$.
5. Записать функцию $f(x,y) = \left(\frac{\mathrm{arctg}(x+y)}{\mathrm{arctg}(x-y)}\right)^2$ в виде функционального оператора и вычислите ее значения при x=1, y=0 и при $x = (1+\sqrt{3})/2$, $y = (1-\sqrt{3})/2$.
5. Записать функцию $f(x,y) = \frac{x^3y^2 - x^2y^3}{(xy)^5}$ с помощью оператора присваивания и вычислите ее значение при x=a, y=1/a, используя команду подстановки **`subs`**.
6. Записать функцию

$$f(x) := \begin{cases} 0 & x \le 0 \\ \sin(x) & 0 < x < \pi \\ 0 & \pi < x \end{cases}$$

Найти ее значение в точке x = 0,456

7. Найти все точные решения системы $\begin{cases} x^2 - 5xy + 6y^2 = 0, \\ x^2 + y^2 = 10. \end{cases}$ в аналитическом виде.

Контрольные вопросы.

1. Что такое *Maple* и для чего он предназначен?
2. Опишите основные элементы окна *Maple*.
3. На какие условные части делится рабочее поле *Maple* и что в этих частях отображается?
4. Как перевести командную строку в текстовую и наоборот?
5. В каком режиме проходит сеанс работы в *Maple*?
6. Перечислите пункты основного меню *Maple* и их назначение.
7. Какое стандартное расширение присваивается файлу рабочего листа *Maple*?
8. Как представляются в *Maple* основные математические константы?
9. Как получить приближенное значение рационального числа?
10. Какими разделительными знаками заканчиваются команды в *Maple* и чем они отличаются?
11. Какой командой осуществляется вызов библиотеки подпрограмм?
12. Объясните назначение команд **`factor`**, **`expand`**, **`normal`**, **`simplify`**, **`combine`**, **`convert`**.
13. Опишите способы задания функций в *Maple*.
14. Какие операции оценивания производятся в *Maple* с действительными выражениями?
15. Для чего предназначена команда **`evalf`**?
16. С помощью каких команд можно найти вещественную и мнимую части комплексного выражения, а также его модуль и аргумент, и комплексно сопряженное ему число? Какую роль выполняет команда **`evalc`**?
17. Для чего предназначена команда **`solve`**?

Часть вторая

Лабораторная работа №2

1. Двумерные графики.
2. Дифференцирование.
3. Интегрирование функции одной переменной.
4. Интегрирование функции многих переменных.
5. Действия с матрицами.

§1. Двумерные графики

Команда `plot` и ее параметры.

Для построения графиков функции **`f(x)`** одной переменной (в интервале $a \le x \le b$ по оси *Ox* и в интервале $c \le y \le d$ по оси *Oy*) используется команда **`plot(f(x), x=a..b, y=c..d, parameters)`**, где **`parameters`** – параметры управления изображением. Если их не указывать, то будут использованы установки по умолчанию. Настройка изображения также может осуществляться с панели инструментов.

Основные параметры команды **`plot`**:

1) **`title="text"`**, где **`text`**-заголовок рисунка (текст можно оставлять без кавычек, если он содержит только латинские буквы без пробелов).
2) **`coords=polar`** – установка полярных координат (по умолчанию установлены декартовы).
3) **`axes`** – установка типа координатных осей: **`axes=NORMAL`** – обычные оси; **`axes=BOXED`** – график в рамке со шкалой; **`axes=FRAME`** – оси с центром в левом нижнем углу рисунка; **`axes=NONE`** – без осей.
4) **`scaling`** – установка масштаба рисунка: **`scaling=CONSTRAINED`** – одинаковый масштаб по осям; **`scaling=UNCONSTRAINED`** – график масштабируется по размерам окна.
5) **`style=LINE(POINT)`** – вывод линиями (или точками).
6) **`numpoints=n`** – число вычисляемых точек графика (по умолчанию **`n=49`**).
7) **`color`** – установка цвета линии: английское название цвета, например, **`yellow`** – желтый и т.д.
8) **`xtickmarks=nx`** и **`ytickmarks=ny`** – число меток по оси *Ox* и оси *Oy*, соответственно.
9) **`thickness=n`**, где **`n=1,2,3`**... - толщина линии (по умолчанию **`n=1`**).
10) **`linestyle=n`** – тип линии: непрерывная, пунктирная и т.д. (**`n=1`** – непрерывная, установлено по умолчанию).
11) **`symbol=s`** – тип символа, которым помечают точки: **`BOX, CROSS, CIRCLE, POINT, DIAMOND`**.
12) **`font=[f,style,size]`** – установка типа шрифта для вывода текста: **`f`** задает название шрифтов: **`TIMES, COURIER, HELVETICA, SYMBOL;`** **`style`** задает стиль шрифта: **`BOLD, ITALIC, UNDERLINE;`** **`size`** – размер шрифта в pt.
13) **`labels=[tx,ty]`** – надписи по осям координат: **`tx`** – по оси *Ox* и **`ty`** – по оси *Oy*.
14) **`discont=true`** – указание для построения бесконечных разрывов.

С помощью команды **`plot`** можно строить помимо графиков функций **`y=f(x)`**, заданной явно, также графики функций, заданных параметрически **`y=y(t), x=x(t)`**, если записать команду **`plot([y=y(t), x=x(t), t=a..b], parameters)`**.

Задание 1.

1. Построить график $y=\frac{\sin x}{x}$ жирной линией в интервале от -4π до 4π. Наберите:

```
> plot(sin(x)/x, x=-4*Pi..4*Pi, labels=[x,y],
labelfont=[TIMES,ITALIC,12], thickness=2);
```



2. Построить график разрывной функции $y=\frac{x}{x^2-1}$.

```
> plot(x/(x^2-1),x=-3..3,y=-3..3,color=magenta);
```



Замечание: на рисунке автоматически появляются вертикальные асимптоты.

3. Построить график параметрической кривой $y=\sin 2t$, $x=\cos 3t$, $0 \le t \le 2\pi$ в рамке. Наберите:

```
> plot([sin(2*t),cos(3*t),t=0..2*Pi], axes=BOXED, color=blue);
```



4. Построить в полярных координатах график кардиоиды $\rho=1+\cos\varphi$ с названием. Наберите:

```
> plot(1+cos(x), x=0..2*Pi, title="Cardioida", coords=polar,
color=coral, thickness=2);
```



5. Построить два графика на одном рисунке: график функции $y = \ln(3x-1)$ и касательную к нему $y = \frac{3}{2}x - \ln 2$. Наберите:

```
> plot([ln(3*x-1), 3*x/2-ln(2)], x=0..6,
scaling=CONSTRAINED, color=[violet,gold],
linestyle=[1,2], thickness=[3,2]);
```



§2. Дифференцирование

Вычисление производных.

Для вычисления производных в *Maple* имеются две команды:

1) прямого исполнения – **diff(f,x)**, где **f** – функция, которую следует продифференцировать, **x** – имя переменной, по которой производится дифференцирование.

2) отложенного исполнения – **Diff(f,x)**, где параметры команды такие же, как и в предыдущей. Действие этой команды сводится к аналитической записи производной в виде $\frac{\partial}{\partial x} f(x)$. После выполнения дифференцирования, полученное выражение желательно упростить. Для этого следует использовать команды **simplify factor** или **expand**, в зависимости от того, в каком виде вам нужен результат.

Пример:

```
> Diff(sin(x^2),x)=diff(sin(x^2),x);
```

$$\frac{\partial}{\partial x}\sin(x^2) = 2\cos(x^2)x$$

Для вычисления производных старших порядков следует указать в параметрах **x$n**, где **n** – порядок производной; например:

```
> Diff(cos(2*x)^2,x$4)=diff(cos(2*x)^2,x$4);
```

$$\frac{\partial^4}{\partial x^4}\cos(2x)^2 = -128\sin(2x)^2 + 128\cos(2x)^2$$

Полученное выражение можно упростить двумя способами:

```
> simplify(%);
```

$$\frac{\partial^4}{\partial x^4}\cos(2x)^2 = 256\cos(2x)^2 - 128$$

```
> combine(%);
```

$$\frac{\partial^4}{\partial x^4}\left(\frac{1}{2}\cos(4x) + \frac{1}{2}\right)^2 = 128\cos(4x)$$

Большинство задач дифференциального исчисления функций многих переменных решается в *Maple* теми же командами, что и для функций одной переменной, только с указанием дополнительных параметров.

Частные производные.

Для вычисления частных производных функции $f(x_1,\ldots,x_m)$ используется уже хорошо известная вам команда **diff**. В этом случае эта команда имеет такой формат: **diff(f,x1$n1,x2$n2,…, xm$nm)**, где **x1,…, xm** – переменные, по которым производится дифференцирование, а после знака **$** указаны соответствующие порядки дифференцирования. Например, частная производная $\frac{\partial^2 f}{\partial x \partial y}$ записывается в виде: **diff(f,x,y)**.

Задание 2.

1. Найти $\frac{\partial f}{\partial x}$ и $\frac{\partial f}{\partial y}$ функции $f = \operatorname{arctg}\frac{x}{y}$.

```
> f:=arctan(x/y):
>D iff(f,x)=simplify(diff(f,x));
```

$$\frac{\partial}{\partial x}\arctan\frac{x}{y} = \frac{y}{x^2+y^2}$$

```
> Diff(f,y)=simplify(diff(f,y));
```

$$\frac{\partial}{\partial y}\arctan\frac{x}{y} = -\frac{x}{x^2+y^2}.$$

2. Найти все частные производные 2-го порядка функции $f(x,y) = \frac{x-y}{x+y}$.

```
> restart; f:=(x-y)/(x+y):
> Diff(f,x$2)=simplify(diff(f,x$2));
```

$$\frac{\partial^2}{\partial x^2}\frac{x-y}{x+y} = -4\frac{y}{(x+y)^3}$$

```
> Diff(f,y$2)=simplify(diff(f,y$2));
```

$$\frac{\partial^2}{\partial y^2}\frac{x-y}{x+y} = 4\frac{x}{(x+y)^3}$$

```
> Diff(f,x,y)=diff(f,x,y);
```

$$\frac{\partial^2}{\partial x \partial y}\frac{x-y}{x+y} = 2\frac{x-y}{(x+y)^3}.$$

§3 Интегрирование функции одной переменной.

Аналитическое и численное интегрирование.

Неопределенный интеграл $\int f(x)dx$ вычисляется с помощью 2-х команд:

1) прямого исполнения – **int(f, x)**, где **f** – подынтегральная функция, **x** – переменная интегрирования;

2) отложенного исполнения – **Int(f, x)** – где параметры команды такие же, как и в команде прямого исполнения **int**. Команда **Int** выдает на экран интеграл в аналитическом виде математической формулы.

Для вычисления определенного интеграла $\int\limits_a^b f(x)dx$ в командах **int** и **Int** добавляются пределы интегрирования, например,

```
> Int((1+cos(x))^2, x=0..Pi)=
 int((1+cos(x))^2, x=0..Pi);
```

$$\int\limits_0^{\pi}(1+\cos(x))^2 dx = \frac{3}{2}\pi$$

Если в команде интегрирования добавить опцию **continuous: int(f, x, continuous)**, то *Maple* будет игнорировать любые возможные разрывы подынтегральной функции в диапазоне интегрирования. Это позволяет вычислять несобственные интегралы от неограниченных функций. Несобственные интегралы с бесконечными пределами интегрирования вычисляются, если в параметрах команды **int** указывать, например, **x=0..+infinity.**

Численное интегрирование выполняется командой **evalf(int(f, x=x1..x2), e)**, где **e** – точность вычислений (число знаков после запятой).

<u>Интегралы, зависящие от параметра. Ограничения для параметров.</u>

Если требуется вычислить интеграл, зависящий от параметра, то его значение может зависеть от знака этого параметра или каких-либо других ограничений. Рассмотрим в качестве примера интеграл $\int_0^{+\infty} e^{-ax}dx$, который, как известно из математического анализа, сходится при a>0 и расходится при a<0. Если вычислить его сразу, то получится:

```
> Int(exp(-a*x),x=0..+infinity)=
  int(exp(-a*x),x=0..+infinity);
Definite integration: Can't determine if the integral is convergent.
Need to know the sign of --> a
Will now try indefinite integration and then take limits.
```

$$\int_0^{+\infty} e^{(-ax)}dx = \lim_{x\to\infty} -\frac{e^{(-ax)}-1}{a}.$$

Таким способом интеграл с параметром не вычислить. Для получения явного аналитического результата вычислений следует сделать какие-либо предположения о значении параметров, то есть наложить на них ограничения. Это можно сделать при помощи команды **assume(expr1)**, где **expr1** – неравенство. Дополнительные ограничения вводятся с помощью команды **additionally(expr2)**, где **expr2** – другое неравенство, ограничивающее значение параметра с другой стороны.

После наложения ограничений на параметр *Maple* добавляет к его имени символ (~), например параметр **a**, на который были наложены некоторые ограничения, в сроке вывода будет иметь вид: a~.

Описание наложенных ограничений параметра **a** можно вызвать командой **about(a)**. Пример: наложить ограничения на параметр a такие, что a>-1, a≤3:

```
> assume(a>-1); additionally(a<=3);
> about(a);
Originally a, renamed a~:
  is assumed to be: RealRange(Open(-1),3)
```

Вернемся к вычислению интеграла с параметром $\int_0^{+\infty} e^{-ax}dx$, которое следует производить в таком порядке:

```
> assume(a>0);
> Int(exp(-a*x),x=0..+infinity)=
  int(exp(-a*x),x=0..+infinity);
```

$$\int_0^{+\infty} e^{(-a\sim x)}dx = \frac{1}{a\sim}$$

Задание 3.

1. Найти неопределенные интегралы: а) $\int \cos x \cos 2x \cos 3x dx$;

б) $\int \frac{3x^4+4}{x^2(x^2+1)^3}dx$.

```
> Int(cos(x)*cos(2*x)*cos(3*x),x)=
  int(cos(x)*cos(2*x)*cos(3*x), x);
```

$$\int \cos(x)\cos(2x)\cos(3x)dx = \frac{1}{8}\sin(2x) + \frac{1}{16}\sin(4x) + \frac{1}{24}\sin(6x) + \frac{1}{4}x$$

```
> Int((3*x^4+4)/(x^2*(x^2+1)^3),x)=
  int((3*x^4+4)/(x^2*(x^2+1)^3),x);
```

$$\int \frac{3x^4+4}{x^2(x^2+1)^3}dx = -4\frac{1}{x} - \frac{57}{8}\arctan(x) - \frac{25}{8}\frac{x}{x^2+1} - \frac{7}{4}\frac{x}{(x^2+1)^2}$$

2. Найти определенный интеграл $\int_0^{\pi/2} \frac{\sin x \cos x dx}{(a^2\cos^2 x + b^2 \sin^2 x)^2}$, при условии a>0, b>0.

```
> assume (a>0); assume (b>0);
> Int(sin(x)*cos(x)/(a^2*cos(x)^2+b^2*sin(x)^2),
  x=0..Pi/2)=int(sin(x)*cos(x)/(a^2*cos(x)^2+b^2*
  sin(x)^2),x=0..Pi/2);
```

$$\int_0^{\pi/2} \frac{\sin(x)\cos(x)}{(a\sim^2\cos(x)^2 + b\sim^2\sin(x)^2)^2}dx = \frac{\ln(b\sim) - \ln(a\sim)}{-a\sim^2 + b\sim^2}$$

3. Найти несобственный интеграл $\int_0^{+\infty} \frac{1-e^{-ax^2}}{xe^{x^2}}dx$, при a>-1

```
> restart; assume(a>-1);
> Int((1-exp(-a*x^2))/(x*exp(x^2)),
  x=0..+infinity)=int((1-exp(-a*x^2))/(x*exp(x^2)),
  x=0..+infinity);
```

$$\int_0^{+\infty} \frac{1-e^{(-a\sim x^2)}}{xe^{x^2}}dx = \frac{1}{2}\ln(a\sim+1)$$

§4 Интегральное исчисление функций многих переменных

В *Maple* имеются две специальные команды для вычисления двойных и тройных интегралов, содержащиеся в библиотеке **student**.

Для вычисления двойных интегралов $\iint_D f(x,y)dxdy$ используется команда **Doubleint(f(x, y), D)**, где **D** – область интегрирования, записываемая в одном из следующих форматов:

- **x=x1..x2, y=y1..y2**, где числа **x1, x2, y1, y2** задают прямоугольную область интегрирования;
- **x=f1(y)..f2(y), y=y1..y2**, где **f1(y), f2(y)** – линии, ограничивающие область интегрирования слева и справа на интервале от **y1** до **y2**;
- **x=x1..x2, y=g1(x)..g2(x)** , где **g1(y), g2(y)** – линии, ограничивающие область интегрирования снизу и сверху на интервале от **x1** до **x2**.

Для вычисления тройных интегралов $\iiint_V f(x,y,z)dxdydz$ используется команда **Tripleint(f(x, y, z),x, y, z, V)**, где **V** – область интегрирования.

Обе эти команды являются командами отложенного действия. Чтобы получить значение интеграла, следует использовать команду **value(%)**.

Повторные интегралы можно вычислять с помощью повторения команды **int**, например, повторный интеграл $\int_0^2 dy \int_0^1 x^2y^3dx$ вычисляется командой

```
> int(int(x^2*y^3, x=0..1), y=0..2);
```

$$\frac{4}{3}$$

Задание 4.

1. Вычислить повторный интеграл $\int_2^4 dy \int_0^y \frac{y^3}{x^2+y^2}dx$

```
> Int(Int(y^3/(x^2+y^2),x=0..y),y=2..4)=
  int(int(y^3/(x^2+y^2), x=0..y),y=2..4);
```

$$\int_2^4 dy \int_0^y \frac{y^3}{x^2+y^2}dx = \frac{14}{3}\pi$$

2. Вычислить двойной интеграл $\iint_D \sin(x+2y)dxdy$ по области, ограниченной линиями $y=0,\ y=x,\ x+y=\frac{\pi}{2}$.

Замечание: сначала следует описать область интегрирования D в виде неравенств: $D=\{(x,y): y \le x \le \frac{\pi}{2}-y,\ 0 \le y \le \frac{\pi}{2}\}$

```
> restart: with(student):
> J:=Doubleint(sin(x+2*y), x=y..Pi/2-y, y=0..Pi/2);
```

$$J := \int_0^{\frac{1}{2}\pi} \int_y^{\frac{1}{2}\pi-y} \sin(x+2y)dxdy$$

```
> J:=value(%);
```

$$J := \frac{2}{3}$$

3. Вычислить тройной интеграл $\int_{-1}^1 dx \int_{x^2}^1 dy \int_0^2 (4+z)dz$.

Замечание: следует помнить, что порядок интегрирования определяется последовательностью пределов, поэтому сначала указываются пределы, содержащие функции.

```
> J:=Tripleint(4+z, y=x^2..1,x=-1..1, z=0..2);
```

$$J := \int_0^2 \int_{-1}^1 \int_{x^2}^1 4+zdydxdz$$

```
> J:=value(%);
```

$$J := \frac{40}{3}$$

§5. Действия с матрицами

Основная часть команд для решения задач линейной алгебры содержится в библиотеке **linalg**. Поэтому перед решением задач с матрицами и векторами следует загрузить эту библиотеку командой **with(linalg)**.

Определение матрицы.

Для определения матрицы в *Maple* можно использовать команду **matrix(n, m, [[a11,a12,…,a1n], [a21,a22,…,a2m],…, [an1,an2,…,anm]])**, где **n** – число строк, **m** – число столбцов в матрице. Эти числа задавать необязательно, а достаточно перечислить элементы матрицы построчно в квадратных скобках через запятую. Например:

```
> A:=matrix([[1,2,3],[-3,-2,-1]]);
```

$$A:=\begin{bmatrix}1 & 2 & 3\\ -3 & -2 & -1\end{bmatrix}$$

Арифметические операции с матрицами.

Сложение двух матриц одинаковой размерности осуществляется теми же командами, что и сложение векторов: **`evalm(A+B)`** или **`matadd(A,B)`**. Произведение двух матриц может быть найдено с помощью двух команд:

1) **`evalm(A&*B);`**
2) **`multiply(A,B).`**

В качестве второго аргумента в командах, вычисляющих произведение, можно указывать вектор, например:

```
> A:=matrix([[1,0],[0,-1]]);
> B:=matrix([[-5,1], [7,4]]);
```

$$A:=\begin{bmatrix}1 & 0\\ 0 & -1\end{bmatrix}\quad B:=\begin{bmatrix}-5 & 1\\ 7 & 4\end{bmatrix}$$

```
> v:=vector([2,4]);
```

$$v:=[2,4]$$

```
> multiply(A,v);
```

$$[2,-4]$$

```
> multiply(A,B);
```

$$\begin{bmatrix}-5 & 1\\ -7 & -4\end{bmatrix}$$

```
> matadd(A,B);
```

$$\begin{bmatrix}-4 & 1\\ 7 & 3\end{bmatrix}$$

Команда **`evalm`** позволяет также прибавлять к матрице число и умножать матрицу на число. Например:

```
> C:=matrix([[1,1],[2,3]]):
> evalm(2+3*C);
```

$$\begin{bmatrix}5 & 3\\ 6 & 11\end{bmatrix}$$

Определители, миноры и алгебраические дополнения. Ранг и след матрицы.

Определитель матрицы A вычисляется командой **`det(A).`** Команда **`minor(A,i,j)`** возвращает матрицу, полученную из исходной матрицы A вычеркиванием i-ой строки и j-ого столбца. Минор M_{ij} элемента a_{ij} матрицы A можно вычислить командой **`det(minor(A,i,j)).`** Ранг матрицы A вычисляется командой **`rank(A).`** След матрицы A, равный сумме ее диагональных элементов, вычисляется командой **`trace(A).`**

```
> K:=matrix([[4,0,5],[0,1,-6],[3,0,4]]);
```

$$A:=\begin{bmatrix}4 & 0 & 5\\ 0 & 1 & -6\\ 3 & 0 & 4\end{bmatrix}$$

```
> det(K);
```

$$1$$

```
> minor(K,3,2);
```

$$\begin{bmatrix}4 & 5\\ 0 & -6\end{bmatrix}$$

```
> det(%);
```

$$-24$$

```
> trace(K);
```

Обратная и транспонированная матрицы.

Обратную матрицу A^{-1} , такую что $A^{-1}A$=AA^{-1}=E, где E – единичная матрица, можно вычислить двумя способами:

1) **`evalm(1/A);`**
2) **`inverse(A).`**

Транспонирование матрицы A – это изменение местами строк и столбцов. Полученная в результате этого матрица называется транспонированной и обозначается A'. Транспонированную матрицу A' можно вычислить командой **`transpose(A).`**

Например, используя заданную в предыдущем пункте матрицу K, найдем ей обратную и транспонированную:

```
> inverse(K);
```

$$\begin{bmatrix} 4 & 0 & -5 \\ -18 & 1 & 24 \\ -3 & 0 & 4 \end{bmatrix}$$

```
> multiply(K,%);
```

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

```
> transpose(K);
```

$$\begin{bmatrix} 4 & 0 & 3 \\ 0 & 1 & 0 \\ 5 & -6 & 4 \end{bmatrix}$$

Задание 5.

1. Даны матрицы: $A=\begin{bmatrix} 4 & 3 \\ 7 & 5 \end{bmatrix}$, $B=\begin{bmatrix} -28 & 93 \\ 38 & -126 \end{bmatrix}$, $C=\begin{bmatrix} 7 & 3 \\ 2 & 1 \end{bmatrix}$. Найти: $(AB)C$, det$A$, det$B$, det$C$, det[$(AB)C$]. Наберите:

```
> restart;
with(linalg): A:=matrix([[4,3],[7,5]]):
> B:=matrix([[-28,93],[38,-126]]):
> C:=matrix([[7,3],[2,1]]):
> F:=evalm(A&*B&*C);
```

$$F=\begin{bmatrix} 2 & 0 \\ 0 & 3 \end{bmatrix}$$

```
> Det(A)=det(A); Det(B)=det(B); Det(C)=det(C);
 Det(F)=det(F);
```

Det(A)=–1
Det(B)=–6
Det(C)=1
Det(F)=6

2. Дана матрица $A=\begin{bmatrix} 2 & 5 & 7 \\ 6 & 3 & 4 \\ 5 & -2 & -3 \end{bmatrix}$, найти: det A, A', A^{-1}, det(M_{22}). Наберите:

```
> A:=matrix ([[2,5,7],[6,3,4],[5,-2,-3]]);
```

$$A:=\begin{bmatrix} 2 & 5 & 7 \\ 6 & 3 & 4 \\ 5 & -2 & -3 \end{bmatrix}$$

```
> Det(A)=det(A);
```

$$\mathrm{Det}(A)=-1$$

```
> transpose(A);
```

$$\begin{bmatrix} 2 & 6 & 5 \\ 5 & 3 & -2 \\ 7 & 4 & -3 \end{bmatrix}$$

```
> inverse(A);
```

$$\begin{bmatrix} 1 & -1 & 1 \\ -38 & 41 & -34 \\ 27 & -29 & 24 \end{bmatrix}$$

```
> det(minor(A,2,2));
```

$$-41$$

Контрольные задания.

1. Построить на отдельных рисунках графики функций Бесселя первого рода $J_n(x)$ для различных ее номеров n в интервале $-20<x<20$. Функции Бесселя вызываются командой **`BesselJ(n,x)`**, где **`n`** – номер функции Бесселя, **`x`** – независимая переменная. Построить первые 6 функций Бесселя для n=0,1,2,3,4,5,6. Как они выглядят и чем отличаются друг от друга? Сделать подписи осей курсивом.
2. Построить график функции $\rho=\cos^3(\varphi/3)$ в полярных координатах при $0<\varphi<4\pi$. Используйте цвет линии под названием **`magenta`**, установите толщину линии 3.
3. Построить график функции
$$f(x) := \begin{cases} 0 & x \le 0 \\ \sin(x) & 0 < x < \pi \\ 0 & \pi < x \end{cases}$$
4. Найти $\dfrac{\partial^5}{\partial x^5}(\ln x)$.
5. Найти все частные производные 2 – ого порядка функции
$$f(x,y)=\mathrm{arctg}\,\frac{x+y}{1-xy}.$$
6. Вычислить неопределенный интеграл $\displaystyle\int\frac{(x^3-6)dx}{x^4+6x^2+8}$.
7. Вычислить несобственный интеграл $\displaystyle\int_0^{+\infty}\frac{\sin(ax)\cos(bx)dx}{x}$ при a>0 b>0 для случаев: 1) $a>b$, 2) $a=b$, 3) $a<b$.
8. Вычислить тройной интеграл:
$$\int_0^{e-1}dx\int_0^{e-x-1}dy\int_e^{x+y+e}\frac{\ln(z-x-y)dz}{(x-e)(x+y-e)}.$$
9. Даны матрицы $A=\begin{bmatrix} 5 & 7 & -3 & -4 \\ 7 & 6 & -4 & -5 \\ 6 & 4 & -3 & -2 \\ 8 & 5 & -6 & -1 \end{bmatrix}$ и $B=\begin{bmatrix} 1 & 2 & 3 & 4 \\ 2 & 3 & 4 & 5 \\ 1 & 3 & 5 & 7 \\ 2 & 4 & 6 & 8 \end{bmatrix}$. Найти: AB, BA, det$A$, det$B$.
10. Дана матрица: $A=\begin{bmatrix} 1 & 2 & 3 & 4 \\ 2 & 3 & 1 & 2 \\ 1 & 1 & 1 & -1 \\ 1 & 0 & -2 & -6 \end{bmatrix}$. Найти: det$A$, A^{-1}, M_{32}, A'.

Контрольные вопросы.

С помощью каких команд строятся графики на плоскости ? Какие аргументы имеют эти команды?
Что такое команды прямого и отложенного исполнения? Опишите их действия.
Какие команды производят аналитическое и численное интегрирование? Опишите их параметры.
С помощью каких команд вводятся ограничения на параметры для вычисления интегралов, зависящих от параметров?
Для чего предназначен пакет **`student`**?
Какой пакет следует загрузить перед решением задач линейной алгебры в *Maple*? Какими двумя командами можно вычислить произведение двух матриц (или матрицы на вектор)?
Какие команды используются для нахождения определителя, минора, алгебраического дополнения, следа матрицы?
Какая матрица называется обратной и какими способами она вычисляется в *Maple*?